

The Instantaneous Trendline

AfTA

20 May 2003

John Ehlers

Philosophical Background

- **Technical Analysts must model the market to attack it**
 - Ellioticians
 - Patterns
 - Fibonacci
- **My models are based on a variant of the Random Walk**
- **My models consist of a Trend Mode and a Cycle Mode**

Diffusion Equation

- **“Drunkard’s Walk” is a special form of the random walk problem**
 - The drunk flips a coin to determine right or left with each step forward
 - The random variable is direction
- **The Diffusion equation is the solution**
 - describes smoke coming from a smokestack
- **The smoke plume is analogous to market conditions**
 - Breeze bends the plume to an average trendline
 - Plume widens with distance - distant predictions are less accurate
 - Smoke density is analogous to prediction probability - the best estimator is the average

Telegrapher's Equation

- **Modify the “Drunkard’s Walk” problem**
 - Coin flip decides whether the drunk will reverse his direction, regardless of the direction of the last step
 - The random variable is now momentum, not direction
- **Solution is now the Telegrapher’s Equation**
 - Describes the electric wave on a telegraph wire
 - Also describes a meandering river
- **A river meander is a short term cycle**
 - Random probability exists (Diffusion Equation) IF:
 - Individual meanders are overlaid
 - Or a long data span is taken

The Market is Similar to a Meandering River

- Both follow the path of least resistance
- Market Forces (greed, fear, profit, loss, etc.) are similar to physical forces, producing paths of uniform resistance.
- Think about how the masses ask the question:
 Will the market change?
 OR
 Will the trend continue?

Establishing Market Modes

- **Historically, I have measured the cycles and subtracted the Cycle components from the total to find the Trend**
- **Tonight, I will parse the two modes directly by filtering**
- **You will also see a successful trading strategy developed**
- **The Trend and Cycle components will be compared**


Exponential Moving Average

In EasyLanguage notation

Output = α *Input + (1 - α)*Output[1];

Using Z Transforms:


$$H(z) = \frac{\text{Output}}{\text{Input}} = \frac{\alpha}{1 - (1 - \alpha) * Z^{-1}}$$



Improved Attenuation

Averaging over 2 samples dramatically improves attenuation

$$H(z) = \frac{\frac{\alpha}{2} * (1 + Z^{-1})}{1 - (1 - \alpha) * Z^{-1}}$$




Moving Average Lag

Lag of Moving Averages is the enemy of technical analysts

$$\alpha = \frac{2}{Length + 1}$$

$$\alpha = \frac{1}{Lag + 1}$$



High Pass Filter Removes the Trend

Create by subtracting EMA Trend from an Allpass (unity)


$$\begin{aligned}
 HP(z) &= 1 - \frac{\frac{\alpha}{2} * (1 + Z^{-1})}{1 - (1 - \alpha) * Z^{-1}} \\
 &= \frac{1 - (1 - \alpha) * Z^{-1} - \frac{\alpha}{2} * (1 + Z^{-1})}{1 - (1 - \alpha) * Z^{-1}} \\
 &= \frac{(1 - \frac{\alpha}{2}) * (1 - Z^{-1})}{1 - (1 - \alpha) * Z^{-1}}
 \end{aligned}$$

Square to improve sharpness

$$HP(z) = \frac{(1 - \frac{\alpha}{2})^2 * (1 - 2 * Z^{-1} + Z^{-2})}{1 - 2 * (1 - \alpha) * Z^{-1} + (1 - \alpha)^2 * Z^{-2}}$$

In EasyLanguage notation:

$$HPF = (1 - \alpha/2)^2 * (Price - 2 * Price[1] + Price[2]) + 2 * (1 - \alpha) * HPF[1] - (1 - \alpha)^2 * HPF[2];$$




Instantaneous Trendline

Created by subtracting the High Pass Filter (Cycle Mode) from an Allpass

Attenuation is similar to an EMA

Skipping the algebra:



$$IT(z) = \frac{(\alpha - \frac{\alpha^2}{4}) + \frac{\alpha^2}{2}Z^{-1} - (\alpha - \frac{3\alpha^2}{4})Z^{-2}}{1 - 2*(1-\alpha)*Z^{-1} + (1-\alpha)^2Z^{-2}}$$




In EasyLanguage:

InstTrend = ($\alpha - (\alpha/2)^2$)*Price + ($\alpha^2/2$)*Price[1] - ($\alpha - 3\alpha^2/4$)*Price[2] + 2*(1- α)*InstTrend[1] - (1- α)²*InstTrend[2];

Zero Frequency Lag is Removed




Instantaneous Trendline - EMA Comparison




Trading Needs a Leading Signal

Add a 2 day momentum to the Instantaneous Trendline to get a 2 bar lead




Momentum Advancing Comes with a Penalty



Nearly a 10 dB filter gain
for a 4 bar cycle

Gain is unimportant to the Instantaneous Trendline
because it has a 25 dB attenuation at a 4 bar cycle



Simple Instantaneous Trendline Strategy

Inputs: Price((H+L)/2),
 alpha(.07);


Vars: Smooth(0),
 ITrend(0),
 Trigger(0);

ITrend = (alpha - alpha*alpha/4)*Price + .5*alpha*alpha*Price[1]
 - (alpha - .75*alpha*alpha)*Price[2] + 2*(1 - alpha)*ITrend[1]
 - (1 - alpha)*(1 - alpha)*ITrend[2];

If currentbar < 7 then ITrend = (Price + 2*Price[1] + Price[2]) / 4;
Trigger = 2*ITrend - ITrend[2];

Plot1(ITrend, "ITrend");
Plot2(Trigger, "Trig");

Strategy Trades on Line Crossings



Practical Strategy Enhancements

- 1) Buy/Sell on a limit after the price has gone against the anticipated position by a factor related to current volatility
- 2) Provide reversal if market has an adverse excursion

Inputs: Price((H+L)/2),
 alpha(.07),
 RngFrac(.35),
 RevPct(1.015);

Vars: Smooth(0),
 ITrend(0),
 Trigger(0);

ITrend = (alpha - alpha*alpha/4)*Price + .5*alpha*alpha*Price[1]
 - (alpha - .75*alpha*alpha)*Price[2] + 2*(1 - alpha)*ITrend[1]
 - (1 - alpha)*(1 - alpha)*ITrend[2];

If currentbar < 7 then ITrend = (Price + 2*Price[1] + Price[2]) / 4;
Trigger = 2*ITrend - ITrend[2];

If Trigger Crosses Over ITrend then Buy at Close - RngFrac*(High - Low) Limit;
If Trigger Crosses Under ITrend then Sell Short at Close + RngFrac*(High - Low) Limit;

If MarketPosition = 1 and Close < EntryPrice/RevPct then Sell Short Next Bar On Open;
If MarketPosition = -1 and Close > RevPct*EntryPrice then Buy Next Bar on Open;


Results Are Comparable to Strategies Costing \$1000 Or More

Future	Net Profit	# Trades	% Profitable	Profit Factor	Max DD
EC (4/81 – 3/03)	\$201,812	230	42.2%	1.89	(\$26,775)
JY (9/81 – 3/03)	\$221,312	229	48.5%	2.50	(\$11,712)
SF (6/76 – 3/03)	\$129,175	337	45.1%	1.52	(\$15,387)

A Low Lag Smoothing Filter

In EasyLanguage notation:

Smooth = (Price + 2*Price[1] + 2*Price[2] + Price[3]) / 6;



Used to remove the high frequency content of the High Pass Filter

A Simple Cycle Indicator

Inputs: Price((H+L)/2),
 alpha(.07);

Vars: Smooth(0),
 Cycle(0);

Smooth = (Price + 2*Price[1] + 2*Price[2] + Price[3])/6;
Cycle = (1 - .5*alpha)*(1 - .5*alpha)*(Smooth - 2*Smooth[1] + Smooth[2])
 + 2*(1 - alpha)*Cycle[1] - (1 - alpha)*(1 - alpha)*Cycle[2];
If currentbar < 7 then Cycle = (Price - 2*Price[1] + Price[2]) / 4;

Plot1(Cycle, "Cycle");
Plot2(Cycle[1], "Trigger");

**The trading indicator is when
the Cycle Component crosses itself delayed by one bar**

Cycle and Trend Mode Indicators are Complementary

A low lag smoothing line crossed the Instantaneous Trendline at the same time the Cycle crosses zero - THIS IS A FIRST!



SUMMARY

- **You have learned how to create an Instantaneous Trendline (Zero Lag)**
- **You have a trend-following trading strategy comparable to commercial strategies**
- **You have an excellent Cycle Mode indicator (an oscillator)**
- **You understand how profound a philosophical model of the market can be**